

# BitBake User Manual

BitBake Team

---

# BitBake User Manual

by BitBake Team

Copyright © 2004 Chris LarsonPhil Blundell

This work is licensed under the Creative Commons Attribution License. To view a copy of this license, visit <http://creativecommons.org/licenses/by/2.0/> or send a letter to Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

# Table of Contents

|                                                                  |   |
|------------------------------------------------------------------|---|
| 1. Introduction .....                                            | 1 |
| Overview .....                                                   | 1 |
| Background and Goals .....                                       | 1 |
| 2. Metadata .....                                                | 2 |
| Description .....                                                | 2 |
| Basic variable setting .....                                     | 2 |
| Variable expansion .....                                         | 2 |
| Immediate variable expansion (:=) .....                          | 2 |
| Appending (+) and prepending (=+) .....                          | 2 |
| Conditional metadata set .....                                   | 3 |
| Conditional appending .....                                      | 3 |
| Inclusion .....                                                  | 3 |
| Python variable expansion .....                                  | 3 |
| Defining executable metadata .....                               | 3 |
| Defining python functions into the global python namespace ..... | 4 |
| Inheritance .....                                                | 4 |
| Tasks .....                                                      | 4 |
| Events .....                                                     | 4 |
| Parsing .....                                                    | 5 |
| Configuration Files .....                                        | 5 |
| Classes .....                                                    | 5 |
| .bb Files .....                                                  | 5 |
| 3. Commands .....                                                | 6 |
| bbread .....                                                     | 6 |
| bitbake .....                                                    | 6 |
| Introduction .....                                               | 6 |
| Usage and Syntax .....                                           | 6 |
| Metadata .....                                                   | 7 |

# List of Examples

|                                                 |   |
|-------------------------------------------------|---|
| 3.1. Executing a task against a single .bb      | 6 |
| 3.2. Executing tasks against a set of .bb files | 7 |
| 3.3. Setting BBFILES                            | 7 |
| 3.4. Depending on another .bb                   | 8 |
| 3.5. Using PROVIDES                             | 8 |
| 3.6. Specifying version preference              | 8 |
| 3.7. Using “bbfile collections”                 | 9 |

# Chapter 1. Introduction

## Overview

BitBake is, at its simplest, a tool for executing tasks and managing metadata. As such, its similarities to GNU make and other build tools are readily apparent. It was inspired by Portage, the package management system used by the Gentoo Linux distribution. BitBake is the basis of the OpenEmbedded [<http://www.openembedded.org/>] project, which is being used to build and maintain a number of embedded Linux distributions, including OpenZaurus and Familiar.

## Background and Goals

Prior to BitBake, no other build tool adequately met the needs of an aspiring embedded Linux distribution. All of the buildsystems used by traditional desktop Linux distributions lacked important functionality, and none of the ad-hoc *buildroot* systems, prevalent in the embedded space, were scalable or maintainable.

Some important goals for BitBake were:

- Handle crosscompilation.
- Handle interpackage dependencies (build time on target architecture, build time on native architecture, and runtime).
- Support running any number of tasks within a given package, including, but not limited to, fetching upstream sources, unpacking them, patching them, configuring them, et cetera.
- Must be linux distribution agnostic (both build and target).
- Must be architecture agnostic
- Must support multiple build and target operating systems (including cygwin, the BSDs, etc).
- Must be able to be self contained, rather than tightly integrated into the build machine's root filesystem.
- There must be a way to handle conditional metadata (on target architecture, operating system, distribution, machine).
- It must be easy for the person using the tools to supply their own local metadata and packages to operate against.
- Must make it easy to collaborate between multiple projects using BitBake for their builds.
- Should provide an inheritance mechanism to share common metadata between many packages.
- Et cetera...

BitBake satisfies all these and many more. Flexibility and power have always been the priorities. It is highly extensible, supporting embedded Python code and execution of any arbitrary tasks.

# Chapter 2. Metadata

## Description

BitBake metadata can be classified into 3 major areas:

- Configuration Files
- .bb Files
- Classes

What follows are a large number of examples of BitBake metadata. Any syntax which isn't supported in any of the aforementioned areas will be documented as such.

## Basic variable setting

```
VARIABLE = "value"
```

In this example, VARIABLE is value.

## Variable expansion

BitBake supports variables referencing one another's contents using a syntax which is similar to shell scripting

```
A = "aval"
B = "pre${A}post"
```

This results in A containing aval and B containing preavalpost.

## Immediate variable expansion (:=)

:= results in a variable's contents being expanded immediately, rather than when the variable is actually used.

```
T = "123"
A := "${B} ${A} test ${T}"
T = "456"
B = "${T} bval"
```

```
C = "cval"
C := "${C}append"
```

In that example, A would contain test 123, B would contain 456 bval, and C would be cvalappend.

## Appending (+=) and prepending (+=)

```
B = "bval"
B += "additionaldata"
```

```
C = "cval"
C += "test"
```

In this example, B is now `bval additionaldata` and C is `test cval`.

## Conditional metadata set

OVERRIDES is a “:” separated variable containing each item you want to satisfy conditions. So, if you have a variable which is conditional on “arm”, and “arm” is in OVERRIDES, then the “arm” specific version of the variable is used rather than the non-conditional version. Example:

```
OVERRIDES = "architecture:os:machine"
TEST = "defaultvalue"
TEST_os = "osspecificvalue"
TEST_condnotinoverrides = "othercondvalue"
```

In this example, TEST would be `osspecificvalue`, due to the condition “os” being in OVERRIDES.

## Conditional appending

BitBake also supports appending and prepending to variables based on whether something is in OVERRIDES. Example:

```
DEPENDS = "glibc ncurses"
OVERRIDES = "machine:local"
DEPENDS_append_machine = " libmad"
```

In this example, DEPENDS is set to `glibc ncurses libmad`.

## Inclusion

Next, there is the `include` directive, which causes BitBake to parse in whatever file you specify, and insert it at that location, which is not unlike **make**. However, if the path specified on the `include` line is a relative path, BitBake will locate the first one it can find within BBPATH.

## Python variable expansion

```
DATE = "${@time.strftime('%Y%m%d',time.gmtime())}"
```

This would result in the DATE variable containing today’s date.

## Defining executable metadata

*NOTE:* This is only supported in .bb and .bbclass files.

```
do_mytask () {
    echo "Hello, world!"
}
```

This is essentially identical to setting a variable, except that this variable happens to be executable shell code.

```
python do_printdate () {
    import time
    print time.strftime('%Y%m%d', time.gmtime())
}
```

This is the similar to the previous, but flags it as python so that BitBake knows it is python code.

## Defining python functions into the global python namespace

*NOTE:* This is only supported in .bb and .bbclass files.

```
def get_depends(bb, d):
    if bb.data.getVar('SOMECONDITION', d, 1):
        return "dependencywithcond"
    else:
        return "dependency"
```

```
SOMECONDITION = "1"
DEPENDS = "${@get_depends(bb, d)}"
```

This would result in DEPENDS containing dependencywithcond.

## Inheritance

*NOTE:* This is only supported in .bb and .bbclass files.

The `inherit` directive is a means of specifying what classes of functionality your .bb requires. It is a rudimentary form of inheritance. For example, you can easily abstract out the tasks involved in building a package that uses autoconf and automake, and put that into a bbclass for your packages to make use of. A given bbclass is located by searching for classes/filename.oeclass in BBPATH, where filename is what you inherited.

## Tasks

*NOTE:* This is only supported in .bb and .bbclass files.

In BitBake, each step that needs to be run for a given .bb is known as a task. There is a command `addtask` to add new tasks (must be a defined python executable metadata and must start with “do\_”) and describe intertask dependencies.

```
python do_printdate () {
    import time
    print time.strftime('%Y%m%d', time.gmtime())
}
```

```
addtask printdate before do_build
```

This defines the necessary python function and adds it as a task which is now a dependency of do\_build (the default task). If anyone executes the do\_build task, that will result in do\_printdate being run first.

## Events

*NOTE:* This is only supported in .bb and .bbclass files.



BitBake also implements a means of registering event handlers. Events are triggered at certain points during operation, such as, the beginning of operation against a given .bb, the start of a given task, task failure, task success, et cetera. The intent was to make it easy to do things like email notifications on build failure.

*[Insert instructions on how to add event handlers here]*

## Parsing

## Configuration Files

The first of the classifications of metadata in BitBake is configuration metadata. This metadata is global, and therefore affects *all* packages and tasks which are executed. Currently, BitBake has hardcoded knowledge of a single configuration file. It expects to find 'conf/bitbake.conf' somewhere in the user specified BBPATH. That configuration file generally has include directives to pull in any other metadata (generally files specific to architecture, machine, *local* and so on.

Only variable definitions and include directives are allowed in .conf files.

## Classes

BitBake classes are our rudimentary inheritance mechanism. As briefly mentioned in the metadata introduction, they're parsed when an `inherit` directive is encountered, and they are located in `classes/` relative to the `dirs` in BBPATH.

## .bb Files

A BitBake (.bb) file is a logical unit of tasks to be executed. Normally this is a package to be built. Inter-.bb dependencies are obeyed. The files themselves are located via the BBFILES variable, which is set to a space seperated list of .bb files, and does handle wildcards.

# Chapter 3. Commands

## bbread

bbread is a command for displaying BitBake metadata. When run with no arguments, it has the core parse 'conf/bitbake.conf', as located in BBPATH, and displays that. If you supply a file on the commandline, such as a .bb, then it parses that afterwards, using the aforementioned configuration metadata.

## bitbake

### Introduction

bitbake is the primary command in the system. It facilitates executing tasks in a single .bb file, or executing a given task on a set of multiple .bb files, accounting for interdependencies amongst them.

### Usage and Syntax

```
$ bitbake --help
usage: bitbake [options] [package ...]
```

Executes the specified task (default is 'build') for a given set of BitBake files.

It expects that BBFILES is defined, which is a space separated list of files to be executed. BBFILES does support wildcards.

Default BBFILES are the .bb files in the current directory.

options:

|                                    |                                                                                                                                                                                                  |
|------------------------------------|--------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| --version                          | show program's version number and exit                                                                                                                                                           |
| -h, --help                         | show this help message and exit                                                                                                                                                                  |
| -bBUILDFILE, --buildfile=BUILDFILE | execute the task against this .bb file, rather than a package from BBFILES.                                                                                                                      |
| -k, --continue                     | continue as much as possible after an error. While the target that failed, and those that depend on it, cannot be remade, the other dependencies of these targets can be processed all the same. |
| -f, --force                        | force run of specified cmd, regardless of stamp status                                                                                                                                           |
| -cCMD, --cmd=CMD                   | Specify task to execute                                                                                                                                                                          |
| -rFILE, --read=FILE                | read the specified file before bitbake.conf                                                                                                                                                      |
| -v, --verbose                      | output more chit-chat to the terminal                                                                                                                                                            |
| -D, --debug                        | Increase the debug level                                                                                                                                                                         |
| -n, --dry-run                      | don't execute, just go through the motions                                                                                                                                                       |
| -p, --parse-only                   | quit after parsing the BB files (developers only)                                                                                                                                                |
| -d, --disable-psyco                | disable using the psyco just-in-time compiler (not recommended)                                                                                                                                  |
| -s, --show-versions                | show current and preferred versions of all packages                                                                                                                                              |

### Example 3.1. Executing a task against a single .bb

Executing tasks for a single file is relatively simple. You specify the file in question, and bitbake parses it and executes the specified task (or "build" by default). It obeys intertask dependencies when doing so.

“clean” task:

```
$ bitbake -b blah_1.0.bb -c clean
```

“build” task:

```
$ bitbake -b blah_1.0.bb
```

### Example 3.2. Executing tasks against a set of .bb files

There are a number of additional complexities introduced when one wants to manage multiple .bb files. Clearly there needs to be a way to tell bitbake what files are available, and of those, which we want to execute at this time. There also needs to be a way for each .bb to express its dependencies, both for build time and runtime. There must be a way for the user to express their preferences when multiple .bb’s provide the same functionality, or when there are multiple versions of a .bb.

The next section, Metadata, outlines how one goes about specifying such things.

Note that the bitbake command, when not using `--buildfile`, accepts a PROVIDER, not a filename or anything else. By default, a .bb generally PROVIDES its packagename, packagename-version, and packagename-version-revision.

```
$ bitbake blah
```

```
$ bitbake blah-1.0
```

```
$ bitbake blah-1.0-r0
```

```
$ bitbake -c clean blah
```

```
$ bitbake virtual/whatever
```

```
$ bitbake -c clean virtual/whatever
```

## Metadata

As you may have seen in the usage information, or in the information about .bb files, the BBFILES variable is how the bitbake tool locates its files. This variable is a space separated list of files that are available, and supports wildcards.

### Example 3.3. Setting BBFILES

```
BBFILES = "/path/to/bbfiles/*.bb"
```

With regard to dependencies, it expects the .bb to define a `DEPENDS` variable, which contains a space separated list of “package names”, which themselves are the `PN` variable. The `PN` variable is, in general, by default, set to a component of the .bb filename.

### Example 3.4. Depending on another .bb

a.bb:

```
PN = "package-a"
DEPENDS += "package-b"
```

b.bb:

```
PN = "package-b"
```

### Example 3.5. Using PROVIDES

This example shows the usage of the `PROVIDES` variable, which allows a given .bb to specify what functionality it provides.

package1.bb:

```
PROVIDES += "virtual/package"
```

package2.bb:

```
DEPENDS += "virtual/package"
```

package3.bb:

```
PROVIDES += "virtual/package"
```

As you can see, here there are two different .bb’s that provide the same functionality (virtual/package). Clearly, there needs to be a way for the person running bitbake to control which of those providers gets used. There is, indeed, such a way.

The following would go into a .conf file, to select package1:

```
PREFERRED_PROVIDER_virtual/package = "package1"
```

### Example 3.6. Specifying version preference

When there are multiple “versions” of a given package, bitbake defaults to selecting the most recent version, unless otherwise specified. If the .bb in question has a `DEFAULT_PREFERENCE` set lower than the other .bb’s (default is 0), then it will not be selected. This allows the person or persons maintaining the repository of .bb files to specify their preferences for the default selected version. In addition, the user can specify their preferences with regard to version.

If the first .bb is named `a_1.1.bb`, then the `PN` variable will be set to “a”, and the `PV` variable will be set to 1.1.

If we then have an `a_1.2.bb`, bitbake will choose 1.2 by default. However, if we define the following variable in a `.conf` that bitbake parses, we can change that.

```
PREFERRED_VERSION_a = "1.1"
```

### Example 3.7. Using “bbfile collections”

bbfile collections exist to allow the user to have multiple repositories of bbfiles that contain the same exact package. For example, one could easily use them to make one’s own local copy of an upstream repository, but with custom modifications that one does not want upstream. Usage:

```
BBFILES = "/stuff/openembedded/*/*.bb /stuff/openembedded.modified/*/*.bb"
BBFILE_COLLECTIONS = "upstream local"
BBFILE_PATTERN_upstream = "^/stuff/openembedded/"
BBFILE_PATTERN_local = "^/stuff/openembedded.modified/"
BBFILE_PRIORITY_upstream = "5"
BBFILE_PRIORITY_local = "10"
```